# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xeyes
pkgver=1.2.0
pkgrel=1
pkgdesc="Follow the mouse/SHAPE extension X demo"
arch=('x86_64')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libx11' 'libxt' 'libxext' 'libxmu' 'libxrender' 'libxi')
makedepends=('xorg-util-macros' 'libxfixes')
source=(https://xorg.freedesktop.org/archive/individual/app/xeyes-${pkgver}.tar.bz2{,.sig})
sha512sums=('2d1614614b3ba3d1ebaa0e9801a2e159389b94aaacafb962d326b7e6dc712c1eb2c12a50ceb10b3135cf0eee6ae86bbf84e73098deb08e73792f14a60616b4d8'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # "Alan Coopersmith <alan.coopersmith@oracle.com>"

build() {
  cd "xeyes-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "xeyes-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
